TÉCNICO LISBOA

O decaimento $H \to \gamma\gamma$ no Modelo Standard

Jorge C. Romão
Instituto Superior Técnico, Departamento de Física & CFTP
A. Rovisco Pais 1, 1049-001 Lisboa, Portugal

2017

Sumário

- Os vértices
- Os diagramas
- A largura de decaimento
- Os diagramas com fermiões
- Os diagramas com bosões de gauge
- Testes
- Resultados

Os Vértices



f
A_μ
$-ieQ_f\gamma_\mu$
f
f
H
$-i\frac{g}{2}\frac{m_f}{m_W}$
f
W^-_α
H
$igm_W g_{\alpha\beta}$
W^+_β
W^-_α
p
A_μ
$-ie\left[g_{\alpha\beta}(p-k)_\mu + g_{\beta\mu}(k-q)_\alpha + g_{\mu\alpha}(q-p)_\beta\right]$
k
q
W^+_β
$W^\pm_\nu$
A_μ
$-iem_W g_{\mu\nu}$
$G^\mp$
$W^\pm_\mu$
H
$\mp\frac{i}{2}g(k-p)_\mu$
k
p
$G^\mp$
G^+_μ
p
A_μ
$-ie(p-q)_\mu$
q
G^-
G^+_μ
H
$-\frac{i}{2}g\frac{m_H^2}{m_W}$
G^-

Os Vértices $\cdots$



$$-ie^2\left(2g_{\alpha\beta}g_{\mu\nu} - g_{\alpha\mu}g_{\beta\nu} - g_{\alpha\nu}g_{\beta\nu}\right)$$



$$2ie^2 g_{\mu\nu}$$



$$-\frac{i}{2}egg_{\mu\nu}$$



$$\mp iep_\mu$$



$$-\frac{i}{2}gm_W$$



$$i\frac{g}{\cos\theta_W}m_Z g_{\alpha\beta}$$

O Ficheiro de Input para QGRAF

Para este problema é muito útil o programa `QGRAF` de Paulo Nogueira, http://cfif.ist.utl.pt/~paulo/. O ficheiro de input é:

```
output= 'HGG.lista' ;

style= 'Styles/sum.sty' ;

model= 'Models/gws-tHooftFeynmanGauge';

in= H;

out=A,A;

loops= 1;

loop_momentum= ;

options= onepi ;
```

O Ficheiro do Modelo para QGRAF

O ficheiro do Modelo Padrão na gauge de Feynman-'tHooft é:

```
* --------------------------------------------------*
* Last Update: 13.05.2008                           *
* --------------------------------------------------*
*                 Higgs

  [H,H,+]

*   electron, muon, tau

  [e1,E1,-]
  [e2,E2,-]
  [e3,E3,-]
```

··· (Ver programa completo em http://porthos.ist.utl.pt/CTQFT)

```
* Gauge Goldstone

  [GWP,WM,A]
  [WP,GWM,A]
  [GWP,GWM,A]
  [GWP,WM,Z]
  [WP,GWM,Z]
  [GWP,GWM,Z]
```

...

QGRAF: Output


```
# file generated by qgraf-3.1

 +(1)*
 prop(WM(1,-k1+p1),WP(2,k1-p1))*
 prop(WM(3,-k1),WP(4,k1))*
 vrtx(WP(2,k1-p1),WM(3,-k1),H(-1,p1))*
 vrtx(WP(4,k1),WM(1,-k1+p1),A(-2,-q1),A(-4,-q2))

 +(1)*
 prop(GWM(1,-k1+p1),GWP(2,k1-p1))*
 prop(GWM(3,-k1),GWP(4,k1))*
 vrtx(GWP(2,k1-p1),GWM(3,-k1),H(-1,p1))*
 vrtx(GWP(4,k1),GWM(1,-k1+p1),A(-2,-q1),A(-4,-q2))
```

··· (Ver output completo em http://porthos.ist.utl.pt/CTQFT)

```
 -(1)*
 prop(cWM(1,k1),CWM(2,-k1))*
 prop(cWM(3,k1-p1),CWM(4,-k1+p1))*
 prop(cWM(5,k1-q1),CWM(6,-k1+q1))*
 vrtx(CWM(2,-k1),cWM(3,k1-p1),H(-1,p1))*
 vrtx(CWM(6,-k1+q1),cWM(1,k1),A(-2,-q1))*
 vrtx(CWM(4,-k1+p1),cWM(5,k1-q1),A(-4,-q2))
 ;
# end
```

...

Os Diagramas

- Com Fermiões



F_1



F_{1a}

- Com Bosões de Gauge



G_1



G_{1a}



G_2



G_{2a}



G_3



G_{3a}



G_4



G_{4a}

TÉCNICO LISBOA

Os Diagramas · · ·



As Amplitudes

Convenções:

- $V_S^\alpha(p,k) = (p-k)^\alpha$
- $W^{-\alpha}(p), W^{+\beta}(k), A^\mu(q)$ com momentos a entrar no vértice:
  $V_G^{\alpha\beta\mu}(p,k,q) = [g_{\alpha\beta}(p-k)_\mu + g_{\beta\mu}(k-q)_\alpha + g_{\mu\alpha}(q-p)_\beta]$
- Para simplificar as expressões omitimos os denominadores dos propagadores.

Obtemos:

$$F_1 = (-ieQ_f)^2(-i\frac{g}{2}\frac{m_f}{m_W})i^3(-1)\mathrm{Tr}[(\not k + m_f)(\not k - \not p + m_f)\gamma_\nu(\not k - \not q_1 + m_f)\gamma_\mu]$$

$$F_{1a} = (-ieQ_f)^2(-i\frac{g}{2}\frac{m_f}{m_W})i^3(-1)\mathrm{Tr}[(\not k + m_f)\gamma_\mu(\not k + \not q_1 + m_f)\gamma_\nu(\not k + \not p + m_f)]$$

$$G_1 = igm_W(-ie)^2(-i)^3 V_{G\beta\alpha}{}^{\mu}(-k+q_1,k,-q_1)V_G^{\alpha\beta\nu}(-k+q_1+q_2,k-q_1,-q_2)$$

$$G_{1a} = igm_W(-ie)^2(-i)^3 V_{G\alpha\beta}{}^{\mu}(k,-k+q_1,-q_1)V_G^{\beta\alpha\nu}(k-q_1,-k+q_1+q_2,-q_2)$$

$$G_2 = G_{2a} = igm_W(-ie)(-ie)(-i)^2 i m_W^2 g^{\mu\nu}$$

As Amplitudes ...

$$G_3 = -\frac{ig}{2} V_{S\alpha}(-k, q_1+q_2)(-ie)^2 m_W (-i)^2 i V_G^{\alpha\mu\nu}(-k+q_1+q_2, k-q_1, -q_2)$$

$$G_{3a} = \frac{ig}{2} V_{S\alpha}(-k, q_1+q_2)(-ie)^2 m_W (-i)^2 i V_G^{\mu\alpha\nu}(k-q_1, -k+q_1+q_2, -q_2)$$

$$G_4 = \frac{ig}{2} V_{S\alpha}(k-q_1-q_2, q_1+q_2)(-ie)^2 m_W i^3 V_G^{\nu\alpha\mu}(-k+q_1, k, -q_1)$$

$$G_{4a} = -\frac{ig}{2} V_{S\alpha}(k-q_1-q_2, q_1+q_2)(-ie)^2 m_W i^3 V_G^{\alpha\nu\mu}(k, -k+q_1, -q_1)$$

$$G_5 = -\frac{ig}{2}(-ie)^2 m_W (-i) i^2 V_S^{\nu}(-k, q_1+q_2) V_S^{\mu}(k, -k+q_1)$$

$$G_{5a} = \frac{ig}{2}(-ie)^2 m_W (-i) i^2 V_S^{\nu}(-k, q_1+q_2) V_S^{\mu}(-k+q_1, k)$$

$$G_6 = \frac{ig}{2}(-ie)^2 m_W (-i) i^2 V_S^{\mu}(k-q_1-q_2, q_1+q_2) V_S^{\nu}(k-q_1, -k+q_1+q_2)$$

$$G_{6a} = -\frac{ig}{2}(-ie)^2 m_W (-i) i^2 V_S^{\mu}(k-q_1-q_2, q_1+q_2) V_S^{\nu}(-k+q_1+q_2, k-q_1)$$

$$G_7 = G_{7a} = -\frac{ig}{2}\frac{m_H^2}{m_W}(-ie)^2 m_W^2 (i)^2 (-i) g^{\mu\nu}$$

As Amplitudes $\cdots$

$$G_8 = -\frac{ig}{2}\frac{m_H^2}{m_W}(-ie)^2 i^3 V_S^\mu(k,-k+q_1)V_S^\nu(k-q_1,-k+q_1+q_2)$$

$$G_{8a} = -\frac{ig}{2}\frac{m_H^2}{m_W}(-ie)^2 i^3 V_S^\mu(-k+q_1,k)V_S^\nu(-k+q_1+q_2,k-q_1)$$

$$G_9 = -\frac{ig}{2}m_W(-ie)^2 i^3(-1)(k-q_1)^\mu(k-q_1-q_2)^\nu$$

$$G_{9a} = -\frac{ig}{2}m_W(-ie)^2 i^3(-1)(-k)^\mu(-k+q_1)^\nu$$

$$G_{10} = G_{10a} = G_{11} = G_{11a} = -\frac{ieg}{2}(-ie)m_W(-i)ig^{\mu\nu}$$

$$G_{12} = igm_W(-ie^2)(-i)^2 g_{\alpha\beta}\left(2g^{\alpha\beta}g^{\mu\nu} - g^{\alpha\mu}g^{\beta\nu} - g^{\alpha\nu}g^{\beta\mu}\right)$$

$$G_{13} = -\frac{ig}{2}\frac{m_H^2}{m_W}\; i^2\; 2ie^2\; g^{\mu\nu}$$

A Largura de Decaimento

A largura de decaimento escreve-se

$$\Gamma = \frac{1}{16\pi}\frac{1}{m_H}\overline{|M|^2}\ \frac{1}{2}$$

onde o factor $1/2$ se deve a termos duas partículas idênticas no estado final. Devido à invariância de gauge todos os diagramas se podem escrever na forma

$$M_i = \frac{e^2 g}{m_W}\frac{1}{16\pi^2}\left[\epsilon_1(q_1)\cdot\epsilon_2(q_2)\ q_1\cdot q_2 - \epsilon_1(q_1)\cdot q_2\ \epsilon_1(q_2)\cdot q_1\right]\ Q_i^2 X_i$$

Obtemos então

$$\Gamma = \frac{\alpha^2 g^2 m_H^3}{1024\pi^3 m_W^2}\sum_i \left|Q_i^2 X_i\right|^2 = \frac{\alpha^2 G_F m_H^3}{128\sqrt{2}\pi^3}\sum_i \left|Q_i^2 X_i\right|^2$$

onde usámos

$$\sum_{\lambda_1,\lambda_2}\left|\epsilon_1(q_1)\cdot\epsilon_2(q_2)\ q_1\cdot q_2\ - \epsilon_1(q_1)\cdot q_2\ \epsilon_1(q_2)\cdot q_1\right|^2 = 2\ (q_1\cdot q_2)^2 = \frac{1}{2}m_H^4$$

Vamos calcular os diagramas dos fermiões



usando as fórmulas explícitas do Apêndice do meu texto `OneLoop`.
Obtemos para as amplitudes

$$iM_{F_1} = (-ieQ_f)^2(-i\frac{g}{2}\frac{m_f}{m_W})i^3(-1)\int \frac{d^4k}{(2\pi)^4}\frac{\text{Tr}[(\not{k}+m_f)(\not{k}-\not{p}+m_f)\gamma_\nu(\not{k}-\not{q}_1+m_f)\gamma_\mu]}{[k^2-m_f^2][(k-p)^2-m_f^2][(k-q_1)^2-m_f^2]}$$

e

$$iM_{F_2} = (-ieQ_f)^2(-i\frac{g}{2}\frac{m_f}{m_W})i^3(-1)\int \frac{d^4k}{(2\pi)^4}\frac{\text{Tr}[(\not{k}+m_f)\gamma_\mu(\not{k}+\not{q}_1+m_f)\gamma_\nu(\not{k}+\not{p}+m_f)]}{[k^2-m_f^2][(k+p)^2-m_f^2][(k+q_1)^2-m_f^2]}$$

Usando a mudança de variável $k \to -k$ no segundo integral podemos reduzir os dois integrais ao mesmo denominador.

Os Diagramas com Fermiões ...

Fazendo os traços obtemos para a soma dos dois diagramas

$$iM_F = \frac{e^2 g}{m_W} \int \frac{d^4k}{(2\pi)^4} \frac{N_{\mu\nu}\ \epsilon(q_1)^\mu \epsilon(q_2)^\nu}{[k^2 - m_f^2][(k-p)^2 - m_f^2][(k-q_1)^2 - m_f^2]}$$

com

$$N_{\mu\nu} = -4m_f^2 \left[g_{\mu\nu}(m_f^2 - q_1 \cdot q_2) + q_{1\nu} q_{2\mu} + (-4 g_{\mu\alpha} q_{1\nu} + 2 q_{1\alpha} g_{\mu\nu}) k^\alpha + 4 k_\mu k_\nu - k^2 g_{\mu\nu} \right]$$

Usando agora as Eqs (A.61-A.63) com

$$r_1 = -q_1, \quad r_2 = -q_1 - q_2, \quad P = x_1 r_1 + x_2 r_2$$

e

$$C = x_1^2 r_1^2 + x_2^2 r_2^2 + 2 x_1 x_2 r_1 \cdot r_2 + m_f^2 - x_1 r_1^2 - x_2 r_2^2 = m_f^2 + 2 x_2(-1 + x_1 + x_2) q_1 \cdot q_2$$

Obtemos

$$M_F = \frac{e^2 g}{m_W} \frac{1}{16\pi^2} \Big[\epsilon_1(q_1) \cdot \epsilon_2(q_2)(q_1 \cdot q_2) - (\epsilon_1(q_1) \cdot q_2)(\epsilon_2(q_2) \cdot q_1) \Big] Q_f^2 X_F$$

$$X_F = -4m_f^2 \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \frac{4x_2^2 + 4(x_1 - 1)x_2 + 1}{m_f^2 + 2x_2(-1 + x_1 + x_2) q_1 \cdot q_2}$$

Os Diagramas com Fermiões ···

- Notar que a divergência potencial dos termos em $k^\mu k^\nu$ e k^2 em $N^{\mu\nu}$ cancelou. Isto porque não pode haver contratermo para este processo pois não existe vértice ao nível do Lagrangiano.
- Mesmo com uso das expressões explícitas este cálculo está longe de ser fácil.
- A amplitude X_F pode ser expressa em termos de funções elementares com o resultado

$$X_F = -2\tau \left[1 + (1-\tau) f(\tau)\right]$$

com

$$f(\tau) = \begin{cases} \left[\sin^{-1}(\sqrt{1/\tau})\right]^2, & \text{se} \quad \tau \geq 1 \\ -\frac{1}{4}\left[\ln(\eta_+/\eta_-) - i\pi\right]^2, & \text{se} \quad \tau < 1 \end{cases}$$

e

$$\eta_\pm = 1 \pm \sqrt{1-\tau}, \qquad \tau = \frac{4m_f^2}{m_H^2}$$

- No seguimento veremos como utilizar `FeynCalc` para obter o mesmo resultado e também para calcular os outros diagramas.

Os Diagramas com Fermiões com FeynCalc


```
(* Definitions *)
dm[mu_]:=DiracMatrix[mu]
dm[5]:=DiracMatrix[5]
ds[p_]:=DiracSlash[p]
mt[mu_,nu_]:=MetricTensor[mu,nu]
fv[p_,mu_]:=FourVector[p,mu]
epsilon[a_,b_,c_,d_]:=LeviCivita[a,b,c,d]
id[n_]:=IdentityMatrix[n]
sp[p_,q_]:=ScalarProduct[p,q]
li[mu_]:=LorentzIndex[mu]
prop[p_,m_]:=m + ds[p]
PV[p_,mu_]:=PolarizationVector[p,mu]

(* Diagrams F1 e F1a *)

NF1:=Tr[prop[k,mf].prop[k-q1-q2,mf].dm[nu].prop[k-q1,mf].dm[mu]]
NF1a:=Tr[prop[-k,mf].dm[mu].prop[-k+q1,mf].dm[nu].prop[-k+q1+q2,mf]]


ampF:=Contract[(NF1+NF1a) PV[q1,mu] PV[q2,nu] ] \
FeynAmpDenominator[PropagatorDenominator[k,mf], \
PropagatorDenominator[k-q1-q2,mf],PropagatorDenominator[k-q1,mf]]
```

Os Diagramas com Fermiões com FeynCalc


```
(* Simplifications *)
onshell={sp[q1,q1]->0, sp[q2,q2]->0,sp[p,p]->mH^2}
kin1={sp[q1,q2]->mH^2/2}

resF:= (-I/Pi^2) OneLoop[k,ampF] /. onshell
ansF=PaVeReduce[resF]

aux =(Pair[Momentum[q1], Momentum[Polarization[q2, I]]]*
Pair[Momentum[q2], Momentum[Polarization[q1, I]]] -
Pair[Momentum[q1], Momentum[q2]]*Pair[Momentum[Polarization[q1, I]],
Momentum[Polarization[q2, I]]])

ans1=-Coefficient[ansF,aux] mf (-1/2 ) /. kin1

c0 = Coefficient[ans1,C0[0, 0, mH^2, mf^2, mf^2, mf^2],0]
c1 = Coefficient[ans1,C0[0, 0, mH^2, mf^2, mf^2, mf^2],1]
c0 = c0 /. mH->Sqrt[4 mf^2/tau]
c1 = c1 /. mH->Sqrt[4 mf^2/tau]

XF= Simplify[c0 + c1 C0[0, 0, mH^2, mf^2, mf^2, mf^2]]
```

Result:

```
                    2                              2     2     2     2
XF= -2 tau - 4 mf   (tau - 1) C0[0, 0, mH , mf , mf , mf ]
```

Os resultados parecem diferentes mas são de facto iguais. De facto, usando $\tau = 4m_f^2/m_H^2 = 2m_f^2/q_1 \cdot q_2$, obtemos

$$\begin{aligned}
X_F^{(1)} &= -4m_f^2 \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \frac{4x_2^2 + 4(x_1-1)x_2 + 1}{m_f^2 + 2x_2(-1+x_1+x_2) q_1 \cdot q_2} \\
&= -4\tau \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \frac{4x_2(-1+x_1+x_2)+1}{\tau + 4x_2(-1+x_1+x_2)} \\
&= -4\tau \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \left[1 + (1-\tau) \frac{1}{\tau + 4x_2(-1+x_1+x_2)} \right] \\
&= -2\tau + 4(\tau-1)\tau \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \frac{1}{\tau + 4x_2(-1+x_1+x_2)} \\
&= -2\tau - 4(\tau-1) m_f^2 C_0(0,0,m_H^2,m_f^2,m_f^2,m_f^2) = X_F^{(2)}
\end{aligned}$$

pois

$$\begin{aligned}
C_0(0,0,m_H^2,m_f^2,m_f^2,m_f^2) &\equiv -\int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \frac{1}{m_f^2 + 2x_2(-1+x_1+x_2) q_1 \cdot q_2} \\
&= -\tau \frac{1}{m_f^2} \int_0^1 dx_1 \int_0^{1-x_1} dx_2 \, \frac{1}{\tau + 4x_2(-1+x_1+x_2)}
\end{aligned}$$

Os Diagramas com Bosões de Gauge: $G_8 + G_{8a} + G_{13}$

Os diagramas com bosões de gauge $W^\pm$ são mais facilmente calculados na chamada gauge de Feynman-'tHooft, porque os numeradores dos propagadores dos $W^\pm$ só têm o termo $-g^{\mu\nu}$ tal como o fotão em QED. O preço a pagar é um elevado número de diagramas. Para fazer este problema, vamos a partir daqui usar só o `FeynCalc`. Resulta que os diagramas $G_8 + G_{8a} + G_{13}$ são proporcionais à massa do bosão de Higgs e portanto têm que ser invariantes de gauge por si. Usando o código (ver em http://porthos.ist.utl.pt/CTQFT)

```
(*Definitions *)
VScalar[p_,k_,m_]:=fv[p-k,m]

(* Diagrams G8 e G8a *)
numG8:= (1/2 mH^2) VScalar[k,-k+q1,mu] VScalar[k-q1,-k+q1+q2,nu]
numG8a:=(1/2 mH^2) VScalar[-k+q1,k,mu] VScalar[-k+q1+q2,k-q1,nu]

ampG8:=Contract[(numG8+numG8a) PV[q1,mu] PV[q2,nu] ] \
FeynAmpDenominator[PropagatorDenominator[k,mW], \
PropagatorDenominator[k-q1-q2,mW],PropagatorDenominator[k-q1,mW]]

resG8:= (-I/Pi^2) OneLoop[k,ampG8] /. onshell
```

Os Diagramas $G_8 + G_{8a} + G_{13}$


```
(* Diagrams G13 *)

numG13:= - mH^2 mt[mu,nu]


ampG13:=Contract[numG13 PV[q1,mu] PV[q2,nu] ] \
FeynAmpDenominator[PropagatorDenominator[k,mW], \
PropagatorDenominator[k+q1+q2,mW]]

resG13:= (-I/Pi^2) OneLoop[k,ampG13] /. onshell

(* Diagrams G8+G8a+G13 *)

ansG8=PaVeReduce[resG8]
ansG13=PaVeReduce[resG13]
ansG8G13 = -Coefficient[ansG8+ansG13,aux]
TestG8G13:= Simplify[ansG8+ansG13 - (-aux) ansG8G13]
XG8G13:= ansG8G13 /. kin
```

obtemos o resultado

```
                 2               2     2     2     2
XG8G13= 2 + 4 mW   C0[0, 0, mH , mW , mW , mW ]
```

Os Diagramas $G_1 + \cdots + G_7 + G_9 + \cdots + G_{12}$


```
(* Diagram G1+G1a *)
numG1:= (mW^2) WMWPA[-k+q1,k,-q1,b,a,mu] \
WMWPA[-k+q1+q2,k-q1,-q2,a,b,nu]
numG1a:= (mW^2) WMWPA[k,-k+q1,-q1,a,b,mu] \
WMWPA[k-q1,-k+q1+q2,-q2,b,a,nu]

ampG1:=Contract[(numG1+numG1a) PV[q1,mu] PV[q2,nu] ] \
FeynAmpDenominator[PropagatorDenominator[k,mW], \
PropagatorDenominator[k-q1,mW],PropagatorDenominator[k-q1-q2,mW]]

resG1:= (-I/Pi^2) OneLoop[k,ampG1] /. onshell
ansG1=PaVeReduce[resG1]
```

$\cdots$ (Ver programa completo em http://porthos.ist.utl.pt/CTQFT)

```
(* Diagrams G12 and G12a *)
numG12:= (-mW^2) mt[a,b] (2 mt[a,b] mt[mu,nu]- \
         mt[a,mu] mt[b,nu] -mt[a,nu] mt[b,mu])

ampG12:=Contract[numG12 PV[q1,mu] PV[q2,nu] ] \
FeynAmpDenominator[PropagatorDenominator[k,mW], \
PropagatorDenominator[k+q1+q2,mW]]

resG12:= (-I/Pi^2) OneLoop[k,ampG12] /. onshell
ansG12=PaVeReduce[resG12]
```

O Resultado Final para a Amplitude

Obtemos

```
In[3]:= XF

                           2                                  2     2     2     2
Out[3]= -2 tauF - 4 mf  (-1 + tauF) C0[0, 0, mH , mf , mf , mf ]

In[4]:= XG

                              2                                  2     2     2     2
Out[4]= 2 + 3 tauW + 6 mW  (-2 + tauW) C0[0, 0, mH , mW , mW , mW ]
```

com

$$\tau_f = \frac{4m_f^2}{m_H^2}, \quad \tau_W = \frac{4m_W^2}{m_H^2}$$

O Gráfico das Amplitudes



Os Testes: Divergências

Num problema tão complexo como este devemos fazer vários testes para ter a certeza que o resultado está certo. Nestes testes `FeynCalc` é um auxiliar precioso. O primeiro teste é verificar que as divergências cancelam. De facto a maior parte dos diagramas é divergente mas o resultado final não pode ser pois não há contratermo $H\gamma\gamma$ no Lagrangiano. Para isso notemos que o resultado de cada diagrama se pode escrever em termos das funções B_0 e C_0. Destas só a função B_0 é divergente com

$$\boxed{\mathrm{Div}(B_0) = \Delta_\epsilon}$$

```
(* Test that the divergences cancell out *)
div={B0[m1_,m2_,m3_]->Div}
ans=ansG1+ansG2+ansG3+ansG4+ansG5+ansG6+ansG7+ansG9+ansG10+ansG11 \
+ansG12 +ansG8G13 /. kin
TestDivDiag=Function[exp,test=exp /. kin; test=\
Simplify[Coefficient[test /.div,Div]]; test=Coefficient[test,aux2];\
         Coefficient[test,mW,2]]
TestDiv:= Simplify[Coefficient[ans /.div ,Div]]
```

com o resultado

```
  TestDiv= 0
```

Tabela das Divergências

Diagrama	Divergência: Coeficiente de $m_W^2 \Delta_\epsilon \ \epsilon_1(q_1) \cdot \epsilon_2(q_2)$
F1+F1a	0
G1+G1a	9
G2+G2a	0
G3+G3a	$-3/4$
G4+G4a	$-3/4$
G5+G5a	$1/2$
G6+G6a	$1/2$
G7+G7a	0
G9+G9a	$-1/2$
G10+G10a	-1
G11+G11a	-1
G12	-6
G8+G8a+G13	0
Soma	0

Testes: Invariância de Gauge

A amplitude para o processo pode ser escrita na forma

$$M = M_{\mu\nu}\, \epsilon^{\mu}(q_1)\epsilon^{\nu}(q_2)$$

A invariância de gauge requer que

$$q_1^{\mu} M_{\mu\nu} = q_2^{\nu} M_{\mu\nu} = 0$$

ou seja

$$M^{\mu\nu} = (g^{\mu\nu}\; q_1 \cdot q_2 - q_1^{\nu} q_2^{\mu})\, \mathcal{M}$$

Portanto se o resultado do `FeynCalc` se escrever

$$M^{\mu\nu} = g^{\mu\nu} q_1 \cdot q_2 C_1 + q_1^{\nu} q_2^{\mu} C_2$$

devemos ter

$$C_1 + C_2 = 0$$

Nota: Não podem aparecer termos proporcionais a q_1^{μ} ou q_2^{ν} pois $\epsilon_1(q_1) \cdot q_1 = \epsilon_2(q_2) \cdot q_2 = 0$

Testes: Invariância de Gauge · · ·


```
(* Test Gauge Invariance *)

c2 = Simplify[Coefficient[ansG, aux2]/sp[q1,q2] /. kin];
c1 = Simplify[Coefficient[ansG, aux1]];

TestGI:= Simplify[c1+c2]
```

com o output

```
In[1]:= TestGI

Out[1]= 0
```

O coeficiente $C_2 = X_G$ é

```
In[2]:= c2

               2        2        2    2        2              2    2    2    2
          2 (mH  + 6 mW  - 6 mW  (mH  - 2 mW ) C0[0,0,mH ,mW ,mW ,mW ])
Out[2]=-----------------------------------------------------------
                                                2
                                              mH
```

para os diagramas do W.

Testes: Comparação com a Literatura

- **Barroso, Pulido, Romão**, `Nucl. Phys B267 (1986), 509`

$$\begin{aligned} X_F &= -4\left[J_1(0,4/\tau_f) - 4J_2(0,4/\tau_f)\right] \\ X_G &= 4\left[4J_1(0,4/\tau_W) - (6+4/\tau_W)J_2(0,4/\tau_W)\right] \end{aligned}$$

- **Gunion, Haber, Kane, Dawson**, `Higgs Hunter's Guide`

$$\begin{aligned} X_F &= -2\tau_f\left[1 + (1-\tau_f)f(\tau_f)\right] \\ X_G &= 2 + 3\tau_W + 3\tau_W(2-\tau_W)f(\tau_W) \end{aligned}$$

com

$$f(\tau) = \begin{cases} \left[\sin^{-1}(\sqrt{1/\tau})\right]^2, & \text{se} \quad \tau \geq 1 \\ -\frac{1}{4}\left[\ln(\eta_+/\eta_-) - i\pi\right]^2, & \text{se} \quad \tau < 1 \end{cases}$$

$$\eta_\pm = 1 \pm \sqrt{1-\tau}, \quad \tau_f = \frac{4m_f^2}{m_H^2}, \quad \tau_W = \frac{4m_W^2}{m_H^2}$$

Testes: Comparação com a Literatura · · ·

A comparação pode ser feita com o código:

```
(* Comparison with NPB267(1986)509 *)

subJ[m_] := {J1 -> -m^2 C0[0, 0, mH^2, m^2, m^2, m^2],
  J2 -> tau/4 (-1/2 - m^2 C0[0, 0, mH^2, m^2, m^2, m^2])}

XFBPR=Simplify[-4 ( J1 -4 J2) /. subJ[mf]]
XGBPR= Simplify[ 4 (4 J1 -( 6 + 4/tau) J2 )  /. subJ[mW]]

TestXFBPR:=Simplify[XF-XFBPR]
TestXGBPR:=Simplify[XG-XGBPR]
```

```
(* Comparison with Higgs Hunter's Guide *)

subftau[m_]:= {ftau[m] -> -2 m^2 C0[0, 0, mH^2, m^2, m^2, m^2] /tau}

XFHHG=Simplify[-2 tau (1 + (1-tau) ftau[mf]) /. subftau[mf]]
XGHHG= Simplify[ 2 + 3 tau + 3 tau (2-tau) ftau[mW]  /. subftau[mW]]

TestXFHHG:=Simplify[XF-XFHHG]
TestXGHHG:=Simplify[XG-XGHHG]
```

O Resultado Final para $H \to \gamma\gamma$

O resultado final é portanto

$$\boxed{\Gamma = \frac{\alpha^2 G_F m_H^3}{128\sqrt{2}\pi^3} \sum_i \left|Q_i^2 X_i\right|^2}$$

com

$$X_F = -2\tau_f - 4m_f^2(-1+\tau_f)C_0(0,0,m_H^2,m_f^2,m_f^2,m_f^2)$$

$$X_G = 2 + 3\tau_W + 6m_W(-2+\tau_W)C_0(0,0,m_H^2,m_W^2,m_W^2,m_W^2)$$

Branching Ratios

As expressões para as várias larguras são:

- $H \to f\overline{f}$

$$\boxed{\Gamma(H \to f\overline{f}) = \frac{G_F\, m_H m_f^2}{4\pi\sqrt{2}}\, N_c\, \beta^3, \quad \beta = \sqrt{1 - 4m_f^2/m_H^2}}$$

- $H \to WW^*\ (m_W < m_H < 2m_W)$

$$\Gamma(H \to WW^*) = \frac{3G_F^2\, m_W^4 m_H}{16\pi^3}\, F(x,\delta), \quad x = \frac{m_W}{m_H}, \quad \delta = \frac{\Gamma_W}{m_H}$$

$$F(x,\delta) = \int_{2x}^{1+x^2} dy\, \frac{\sqrt{y^2 - 4x^2}}{(1-y)^2 + x^2\delta^2}\, \left(y^2 - 12x^2 y + 8x^2 + 12x^4\right)$$

- $H \to ZZ^*\ (m_Z < m_H < 2m_Z)$

$$\Gamma(H \to ZZ^*) = \frac{G_F^2\, m_Z^4 m_H}{64\pi^3}\left(7 - \frac{40}{3}\sin^2\theta_W + \frac{160}{9}\sin^4\theta_W\right) F(x',\delta'), \ x' = \frac{m_Z}{m_H}, \ \delta' = \frac{\Gamma_Z}{m_H}$$

Branching Ratios · · ·

- $H \to \gamma\gamma$

$$\boxed{\Gamma(H \to \gamma\gamma) = \frac{G_F\, m_H^3}{8\pi\sqrt{2}} \frac{\alpha^2}{16\pi^2} \sum_i |Q_i X_i|^2}$$

- $H \to Z\gamma$

$$\boxed{\Gamma(H \to Z\gamma) = \frac{G_F\, m_H^3}{4\pi\sqrt{2}} \frac{\alpha^2}{16\pi^2} \left(1 - \frac{m_Z^2}{m_H^2}\right)^3 |Y_F + Y_G|^2}$$

$$\boxed{Y_F = \sum_f N_{cf} \frac{Q_f g_V^f}{\sin\theta_W \cos\theta_w} I_F, \quad Y_G = \frac{1}{\tan\theta_W} I_W}$$

com

$$\begin{aligned} I_F \quad = \quad & \frac{8 m_f^2 m_Z^2}{(m_H^2 - m_Z^2)^2} \left[B_0(m_H^2, m_f^2, m_f^2) - B_0(m_Z^2, m_f^2, m_f^2)\right] \\ & - \frac{4 m_F^2}{m_H^2 - m_Z^2} \left[-2 + \left(-4 m_f^2 + m_H^2 - m_Z^2\right) C_0(m_Z^2, 0, m_H^2, m_f^2, m_f^2, m_f^2)\right] \end{aligned}$$

Branching Ratios ...

e

$$\begin{aligned}
I_W = &-\frac{1}{(m_H^2-m_Z^2)^2}\left[m_H^2(1-\tan^2\theta_W)-2m_W^2(-5+\tan^2\theta_W)\right] m_Z^2 \Delta B_0 \\
&-\frac{1}{m_H^2-m_Z^2}\left[m_H^2(1-\tan^2\theta_W)-2m_W^2(-5+\tan^2\theta_W)\right. \\
&\qquad \left. +2m_W^2\left((-5+\tan^2\theta_W)(m_H^2-2M_W^2)-2m_Z^2(-3+\tan^2\theta_W)\right) C_0\right]
\end{aligned}$$

$$\Delta B_0 = B_0(m_H^2, m_W^2, m_W^2) - B_0(m_Z^2, m_W^2, m_W^2), \quad C_0 = C_0(m_Z^2, 0, m_H^2, m_W^2, m_W^2, m_W^2)$$

- $H \to gg$

$$\Gamma(H \to gg) = \frac{\alpha_s^2 G_F m_H^3}{64\pi^3\sqrt{2}} \sum_{i=u,d,s,c,t,b} |X_{F_i})|^2$$

Gráfico Final



Bibliografia

- **Barroso, Pulido, Romão**, `Nucl. Phys. B267 (1986), 509`
- **Gunion, Haber, Kane, Dawson**, `Higgs Hunter's Guide`
- **Página de Métodos Computacionais:** http://porthos.ist.utl.pt/CTQFT
- **QGRAF:** http://cfif.ist.utl.pt/~paulo/
- **Keung, Marciano**, `Phys. Rev. D30 (1984), 248`